

## The Dynamic Duo: ODS Layout and the ODS Destination for PowerPoint

Jane Eslinger, SAS Institute Inc., Cary, North Carolina

### ABSTRACT

Like a good pitcher and catcher in baseball, ODS layout and the ODS destination for PowerPoint are a winning combination in SAS® 9.4. With this dynamic duo, you can go straight from performing data analysis to creating a quality presentation. The ODS destination for PowerPoint produces native PowerPoint files from your output. When you pair it with ODS layout, you are able to dynamically place your output on each slide.

Through code examples this paper shows you how to create a custom title slide, as well as place the desired number of graphs and tables on each slide. Don't be relegated to the sidelines—increase your winning percentage by learning how ODS layout works with the ODS destination for PowerPoint.

### INTRODUCTION

The ODS destination for PowerPoint was first introduced in SAS® 9.4. The destination enables you to create SAS® procedure output in the PresentationML format used by Microsoft PowerPoint 2010 and 2013. As with other ODS destinations, the PowerPoint destination uses table and style templates to render the tables and graphs that you produce using procedures. The syntax for using the ODS destination for PowerPoint is also the same as that for other destinations. You specify the destination and the presentation filename in the ODS statement and close the destination after all desired output has been created:

```
ods powerpoint file ="example.pptx";  
<procedure code>  
ods powerpoint close;
```

The ODS LAYOUT statements have been available since SAS® 9.1. However, they were not fully supported until SAS 9.4. SAS supports two types of layout, but only one is supported for the PowerPoint destination. The type of layout supported for PowerPoint is gridded layout. Gridded layout enables you to arrange output in a two-dimensional grid structure. The basic syntax is as follows:

```
ods layout gridded;  
ods region;  
proc print data=sashelp.baseball;  
run;  
ods layout end;
```

The size of the grid cell is dynamically computed to fit the output object. Within the grid structure, there are a number of different ways to arrange objects, such as placing tables next to graphs, text next to tables, or text above both a table and a graph. ODS LAYOUT statements give you the power and flexibility to arrange the output on each slide.

The LAYOUT= option in the ODS POWERPOINT statement applies a predefined layout. ODS LAYOUT statements perform a different function. They write multiple output objects to one slide in a layout that you specify. LAYOUT= is discussed in the last section of this paper.

Both methods can be used within one presentation, but they should not be used on the same slide. For all examples in this paper, the code uses ODS LAYOUT. It does not specify a predefined layout.

This paper is meant to teach the basic techniques of using ODS LAYOUT with the PowerPoint destination. As you read this paper and work through the examples, recognize that the ODS destination for PowerPoint and ODS for other destinations behave in similar though not identical ways. The PowerPoint destination is designed specifically for presentations. Therefore, it is meant for graphs, short tables, and brief text. The amount of information that can fit on one 8.5x11 page in a PDF file does not necessarily fit on one slide in PowerPoint. The size of the text, tables, and graphs is very large compared to that for all other destinations. Presentations are, by their nature, meant to be viewed on a large screen by an audience.

### THE GAME PLAN—DECIDING WHAT YOU NEED TO DO

Before every game, a pitcher and catcher develop a game plan. They decide how they are going to approach each hitter. You should develop a game plan too. Before you begin programming, you should have a good idea of what you want the presentation to contain and how you want it to look. For example, are you going to have slides that

contain only graphs, only tables, or only text, or some combination of all three? Most likely, you will need all three, which means you need to determine how many objects you want to appear on each slide.

As you begin planning, it is important to consider both how the creation of a new slide is triggered in the destination and the size of your output object. These topics are addressed next.

## TRIGGERING A NEW SLIDE

The default behavior of the PowerPoint destination is to output one object per slide, along with the title and footnote. The purpose of ODS LAYOUT is to output multiple objects in a grid structure on each slide. It therefore overrides the PowerPoint destination default behavior.

There are two ways to trigger a new slide when you are using gridded layout. The first method is to fill the entire slide. When the slide is full, PowerPoint is forced to place the next output object, or container, on the next slide.

The second method for triggering a new slide is to use a DATA \_NULL\_ step between layout containers. The DATA \_NULL\_ method should be used when the layout container does not fill a slide but you need the next layout container to start on the next slide. The code should be placed after the ODS LAYOUT END statement:

```
data _null_;  
  dcl odsout obj();  
run;
```

If you have tried using ODS LAYOUT with PowerPoint before, you might have seen that graphs that you expect to see on the second slide were placed at the bottom of the first slide, and the graph objects were very small. This behavior occurs when the layout container does not fill the first slide. The destination tries to fill the slide by adjusting the size of the output objects. The DATA \_NULL\_ step above should remedy this problem.

Also, be aware that the ODSLST, ODSTEXT, and ODSTABLE procedures all have an option in the procedure statement called PAGEBREAK. When PAGEBREAK=YES, the procedure generates a page break before the procedure is run. Note that the option does not help you generate a new slide after the procedure.

## SETTING OUTPUT OBJECT SIZE

In addition, variable labels, axis and marker labels, and BY line strings all affect the size of your output. Graphs and tables might be removed automatically from the presentation if they do not fit within the layout container due to the size of the output object. If the output does not fit in the size of the layout or region, the destination reverts to its default behavior of one output object per slide. Usually, a warning message is written to the log if either one of these issues occurs.

The easiest way to ensure that your graph or table fits into the grid cell is to let the PowerPoint destination size the output. The PowerPoint destination optimizes the size of the output to fit into the size of the space available. However, when placing multiple graphs on a slide, you should specify the size of a graph as you write the procedure code. For the ODS GRAPHICS procedures, the size is specified in the ODS GRAPHICS statement. For SAS/GRAPH® procedures, the size is specified in the GOPTIONS statement.

## UNDERSTANDING DEFAULT BEHAVIOR

When an ODS layout container is started, all of the output is saved until an ODS LAYOUT END statement is reached. After this statement is reached, the destination begins creating the output. You might encounter strange behavior in titles and footnotes because of this accumulation of output. Titles that belong to the last object in the layout might end up with the first object because everything is exported together. Be sure to use null TITLE and FOOTNOTE statements where needed to avoid this issue. Also, the NOGTITLE and NOGFOOTNOTE options should be placed in the ODS POWERPOINT statement as needed to prevent the titles and footnotes from being part of the graph. Remember, anything that is part of the graph contributes to the amount of space required by the object and might prevent the graph from fitting in the grid cell.

Also, settings from the OPTIONS() option in the ODS POWERPOINT statement can be changed only between slides. If a layout container starts at the bottom of slide 1 but continues to slide 2, those option values remain in effect for the second slide. If your second slide contains a specific background image or color that you do not expect, this is a good indication that the first slide was not full and the layout for the second slide actually started on the first slide.

The ODS destination for PowerPoint is a new technology, so you might encounter a few unexpected quirks or bugs. As with most software, it is best to use the latest version of SAS available.

## THE FIRST INNING—CREATING THE TITLE SLIDE

The first inning of a baseball game is very important. It's when a pitcher and catcher start putting their game plan into action and set the tone for the rest of the game. The title slide of a presentation can play the same role. It establishes the theme and look of the presentation and lets the audience know what to expect.

By design, title slides do not have much information on them. Normally, they contain the presentation title, the names of the authors, and perhaps the date and the meeting title. Therefore, a title slide is not likely to require as much work as other slides to put that information exactly where you want it.


The first step in determining whether you need to adjust the placement of the text on a title slide is to look at what is created by default.

### GENERATING A TITLE SLIDE WITH PROC ODSTEXT

PROC ODSTEXT is very useful for creating a title slide because it outputs text blocks. You can place as many pieces of text within the block as necessary using the P statement, as shown here:

```
title;
footnote;
ods powerpoint file='title1.pptx';
proc odstext;
  p "ODS LAYOUT WITH POWERPOINT IS AWESOME!";
  p "SGF 2016";
  p "Jane Eslinger";
run;
ods powerpoint close;
```


The text from PROC ODSTEXT is placed in the top left corner of the slide by default, as shown below in Display 1.



#### Display 1. Slide Created with the P Statement in PROC ODSTEXT

A slide created using only PROC ODSTEXT is not very appealing. The style templates supplied by SAS for the ODS destination for PowerPoint, `Styles.PowerPointLight` and `Styles.PowerPointDark`, contain special style elements for styling titles and subtitles. The style elements change the font size and color and center the text, though it is still placed at the top of the slide. Here is an example. Display 2 shows the improved look.

```
ods powerpoint file='test2.pptx';
proc odstext;
  p "ODS LAYOUT WITH POWERPOINT IS AWESOME!" / style=PresentationTitle;
  p "SGF 2016" / style=PresentationTitle2;
  p "Jane Eslinger" / style=PresentationTitle2;
run;
ods powerpoint close;
```



## Display 2. PresentationTitle Style Element Applied to P Statement in PROC ODSTEXT

Display 2 looks better than Display 1, but it might not fulfill your needs. Maybe you need the text moved over and down because of your company logo or a conference logo. Maybe you just want it in a different location. This is a perfect situation in which to use ODS LAYOUT.

## CHANGING TITLE PLACEMENT WITH ODS LAYOUT

ODS LAYOUT statements and their options allow you to move a region container, which holds the text, to another location other than the default. The X= option in the ODS LAYOUT GRIDDED statement specifies the horizontal starting position of the layout. The Y= option controls the vertical starting position of the layout. The default location is  $x=0$   $y=0$ , which is the top left corner of the layout container.

For the title slide, you most likely need only one container that is one row and one column wide, which is the default grid that is generated simply by submitting an ODS LAYOUT GRIDDED statement. The container is filled with the text from the PROC ODSTEXT step.

The X= and Y= options are used in the following example to move the text to the right and down, as illustrated in Display 3. This placement is vital to preventing the text from being written over the top of the background image.

```
ods powerpoint file='title3.pptx'
  options(backgroundimage="sgf16_background.jpg");

ods layout gridded x=25pct y=35pct;
  ods region;
  proc odstext;
    p "ODS LAYOUT WITH POWERPOINT IS AWESOME!" / style=PresentationTitle;
    p "SGF 2016" / style=PresentationTitle2;
    p "Jane Eslinger" / style=PresentationTitle2;
  run;
ods layout end;
ods powerpoint close;
```



### Display 3. Layout Container Moves Text Down and to the Right Using the X= and Y= Options

The above code used a percentage to place the layout in the desired position. There are a number of different units of measure that could have been used. You must use one of the units of measure when specifying the x and y position. If you do not, the gridded layout is centered by default.

Notice that the ODS LAYOUT GRIDDED statement does not have any additional arguments—because it does not need any. Only the X= and Y= options are necessary to adjust the placement of the layout container, which is one row long and one column wide. The ODS REGION statement does not require options for this example either.


## THE SECOND INNING—OUTPUTTING GRAPHS

The game plan has been established and the tone has been set with the title slide. It is now time to get into the heart of the presentation to show the world the results of your hard work and analysis. You might want to start your presentation by displaying multiple graphs on one slide. The syntax for placing two graphs on one slide has six key steps, as highlighted below:

```
title "Two Graphs Per Slide";  
ods powerpoint file='twographs1.pptx' nogtitle nogfootnote; ❶  
ods graphics / width=4.5in height=4in; ❷  
ods layout gridded columns=2; ❸  
ods region; ❹  
proc sgplot data=sashelp.baseball;  
  where league = "National";  
  vbar team / response=nruns group=division;  
run;  
ods region; ❺  
proc sgplot data=sashelp.baseball;  
  where league = "American";  
  vbar team / response=nruns group=division;  
run;  
ods layout end; ❻  
ods powerpoint close;
```

- ❶ The NOGTITLE and NOGFOOTNOTE options are added to the ODS POWERPOINT statement to force any titles and footnotes to be part of the slide instead of part of the graph.
- ❷ The WIDTH= and HEIGHT= options set the size of both graphs so that they fit within the region that is half the width of the slide. Without these options, PowerPoint reverts to default behavior and puts one graph per slide.
- ❸ The ODS LAYOUT GRIDDED statement starts the layout container. The COLUMNS= option is set to 2 so that the output is placed side-by-side.
- ❹ The ODS REGION statement begins the region container that is to hold the first graph.
- ❺ The ODS REGION statement begins the region container that is to hold the second graph.
- ❻ The ODS LAYOUT END statement ends the gridded layout.

The resulting slide is shown below, in Display 4.




Display 4. Two Graphs on One Slide

The code shown above used two procedure steps, where each procedure filled one region. It is possible to fill multiple regions using only a single procedure. This is useful when one procedure generates multiple output objects. The ADVANCE= option in the ODS LAYOUT GRIDDED statement specifies how the grid should be populated. It can be explicitly populated or dynamically populated by BY groups, tables, or procedures.

The preceding code can be simplified to use just one SGPLOT procedure with the addition of a BY statement. The ADVANCE= option can be used to put the output in the appropriate regions, as shown here and in Display 5:

```
proc sort data=sashelp.baseball out=baseball;
  by league;
run;
title "Two Graphs Per Slide Using ADVANCE=";
ods powerpoint file='twographs2.pptx' nogtitle nogfootnote;
ods graphics / width=4.5in height=4in;
ods layout gridded columns=2 advance=bygroup;
  ods region;
  proc sgplot data=baseball;
    by league;
    vbar team / response=nruns group=division;
  run;
ods layout end;
ods powerpoint close;
```




Display 5. Two Graphs on One Slide Using ADVANCE=

The number of graphs per slide can be increased to four and displayed in a 2x2 pattern. A slide might also hold more than four graphs, but when you start increasing the number of output objects, their size becomes critically important. And remember that titles, BY lines, and axis labels use valuable space.

As in the previous example, this next example uses `COLUMNS=2` and `ADVANCE=BYGROUP` in the ODS LAYOUT GRIDDED statement. The `ROWS=` option is added and set to 2 so that each slide has two columns and two rows creating a 2x2 pattern. The `WIDTH=` and `HEIGHT=` values in the ODS GRAPHICS statement have changed because the graphs need to be much smaller.

The SASHELP.BASEBALL data set contains data for both leagues and both divisions within the league, which is perfect for generating four graphs, as shown in Display 6. Both the LEAGUE and DIVISION variables are in the BY statement.

```
proc sort data=sashelp.baseball out=baseball;
  by league division;
run;
ods powerpoint file='twobytwol.pptx' nogtitle nogfootnote;
title "Graphs in a 2x2 Pattern";
ods graphics on / height=2.25in width=3in;
ods layout gridded columns=2 rows=2 advance=bygroup;
ods region;
proc sgplot data=baseball;
  by league division;
  vbar team / response=nruns;
  label league='League' division='Division';
run;
ods layout end;
ods powerpoint close;
```



**Display 6. Four Graphs on One Slide Using COLUMNS=2 and ROWS=2**

Notice that a LABEL statement is included in the code above. The labels for these variables are both very long. The original full text of the BY line was “League at the End of 1986=American Division at the End of 1986=East.” Using this text would have generated the following warning in the log, and only the first row of graphs would have fit on the slide. The labels were shortened to take up less room so that all four graphs fit on the slide together.

WARNING: GRIDDED Layout exceeds the available space for POWERPOINT destination and will be paneled.

Note that the example above had `ROWS=2` and `COLUMNS=2`. Specifying both `ROWS=` and `COLUMNS=` values in the ODS LAYOUT GRIDDED statement hardcodes the number of cells in the layout. This technique is acceptable when you know you are outputting only four graphs. However, if the data changes or you know that you need to output more than four graphs, hardcoding the number of cells is likely to cause problems. The following warnings are written to the log if the procedure outputs more graphs than the number of available cells. Also, the first slide will contain four graphs, but all other slides will contain just one graph.

```
WARNING: LAYOUT is full. Implicitly closing LAYOUT.
WARNING: LAYOUT is full. Implicitly closing LAYOUT.
NOTE: There were 322 observations read from the data set WORK.BASEBALL.

103 ods layout end;
WARNING: Missing LAYOUT_START statement. This statement will be ignored.
WARNING: Missing LAYOUT_START statement. This statement will be ignored.
```

The output in Display 6, above, can be generated without hardcoding the number of cells in the layout. The following code demonstrates specifying the region height to place four graphs per slide:

```
proc sort data=sashelp.baseball out=baseball;
  by league division;
run;
ods powerpoint file='twobytwo2.pptx' nogtitle nogfootnote;
title "Graphs in a 2x2 Pattern";
ods graphics on / height=2.25in width=3in; ❶
```

*code continues*



```
ods layout gridded columns=2 advance=bygroup; ❷
ods region height=47%; ❸
proc sgplot data=baseball;
  by league division;
  vbar team / response=nruns;
  label league='League' division='Division'; ❹
run;
ods layout end;
ods powerpoint close;
```

- ❶ The height and width of the graph are specified to ensure that the graph fits within the region cell.
- ❷ Only one dimension of the gridded layout is specified, **COLUMNS=2**.
- ❸ The **HEIGHT=** option in the ODS REGION statement specifies that the region should use approximately half of the height of the slide.
- ❹ The labels for the LEAGUE and DIVISION are shortened, so that the graphs fit in the region.

You might wonder why the height of the regions was specified in the ODS REGION statement instead of in the ODS LAYOUT GRIDDED statement using the ROW\_HEIGHTS= option. If ROW\_HEIGHTS= is used and set to 47%, only two graphs are output and a warning is generated for each output object that does not appear in the presentation.

WARNING: The region or layout was too small to accommodate the output supplied for the POWERPOINT destination. Output will be suppressed.

The ROW\_HEIGHTS= option is for the height of each row specified. The ODS LAYOUT GRIDDED statement does not contain the ROWS= option, which means it defaults to a value of **one** (1). All four graphs cannot fit into a row that is 47% of the available space, so PowerPoint suppresses two of the graphs and issues the warning.

The SAS® 9.4 Output Delivery System documentation describes the calculation of row heights this way:

*If omitted, the ROWS= option defaults to the maximum number of rows needed to populate the regions created in the vertical direction. If there are two columns, then the number of rows is half of the number of regions.*

When ROWS= is not used, it is better to specify the HEIGHT= option in the ODS REGION statement, which sets the height of each region.

## THE MIDDLE INNINGS—SHOWING OFF THE NUMBERS

The pitcher and catcher established a game plan and set it into motion with the title slide and the most important graphs. The middle part of the game or presentation is for backing up your claims. You have to show the most important data values, just like a pitcher proving his stuff can last the entire game.

Data values are often output in tabular form. In general, tables that are placed in a presentation should be wider rather than longer. A long table might span multiple slides, making it difficult for the audience to read. Table output objects present size and spacing challenges just like graphs. You have to pay attention to the number of columns, number of rows, labels, and width of the data values themselves. Often, instead of outputting the data in one large table, like you would in the PDF or HTML destinations, it is best to output the data in multiple smaller tables.

The SASHELP.BASEBALL data set contains two variables, NHITS and NRUNS, which are of great interest and need to be included in the presentation. The count, mean, minimum, and standard deviation are key statistics to display in the table. The goal of this example is to generate the N, MEAN, MIN, and STD statistics for the NHITS and NRUNS variables within League and Division and to have it all fit on one slide.

### PLACE TWO TABLES ON ONE SLIDE

Several different procedures could be used to generate the needed statistics for these two variables. For this paper, the REPORT procedure was chosen over the MEANS or TABULATE procedures because it allows more control over the table structure. This flexibility makes it easier to fit the tables inside the layout container. You must choose which procedure works best for your presentation.

The next example takes the preemptive step of generating a separate table for each variable in an attempt to prevent

the table output object from spanning multiple slides. As you will see, it often requires multiple attempts to get the desired output. As shown below, the ODS LAYOUT GRIDDED statement uses `ROWS=2` so that the tables are stacked on top of each other. The results of this code are in Display 7.

```
ods powerpoint file='tables1.pptx' nogtitle nogfootnote;
title 'Data By League and Division';
ods layout gridded rows=2;
ods region;
proc report data=sashelp.baseball;
  column league division nruns,(n mean min std);
  define league / group 'League';
  define division / group 'Division';
run;
ods region;
proc report data=sashelp.baseball;
  column league division nhits,(n mean min std);
  define league / group 'League';
  define division / group 'Division';
run;
ods layout end;
ods powerpoint close;
```

The image shows two stacked PowerPoint slides, each with a title and a table. The top slide is titled "Data By League and Division" and contains a table for "Runs in 1986". The bottom slide is also titled "Data By League and Division" and contains a table for "Hits in 1986".

		Runs in 1986			
League	Division	n	mean	min	std
American	East	85	58.517647	15	28.099149
	West	90	53.188889	13	23.929623
National	East	72	49.666667	12	23.265159
	West	75	46.36	14	23.05378

		Hits in 1986			
League	Division	n	mean	min	std
American	East	85	113.32941	36	48.772403
	West	90	102.35556	39	41.168689

		Hits in 1986			
League	Division	n	mean	min	std
National	East	72	99.805556	31	43.805472
	West	75	96.84	32	41.427253

Display 7. Two Tables in a Two-Row Gridded Layout

Display 7, above, shows that the results of the two PROC REPORT steps and the `ROWS=2` option span multiple slides, even after you put each analysis variable in its own table and carefully choose the procedure. The code placed no height or width restrictions on the regions. Therefore, the destination determined the size of the output objects, and a portion of the second table is placed on a second slide.

You might consider using the `ROW_HEIGHTS=` option in the ODS LAYOUT GRIDDED statement. This will result in the output being limited to one slide. However, the two data rows on the second slide in Display 7 will be suppressed, which is not a desired result.

Both tables contain six rows, which does not seem tall, but remember, the PowerPoint destination uses a large font and contains a title. The slide is just not big enough to hold a title and 12 rows of data. In circumstances like this, you have to rethink how to output the data because both tables have to fit on one slide. Otherwise, it throws off the rest of the presentation. Sometimes the answer to getting output objects to fit inside a gridded layout is to change the output objects and not the grid.

The tables from Display 7 can be modified to remove one header row each. The labels for the variables can be placed inside an ODS TEXT= statement, which can also serve as a title. The changes are demonstrated in the code below and depicted in Display 8.

```
ods powerpoint file='tables2.pptx' nogtitle nogfootnote;
title; ❶
ods layout gridded rows=2; ❷
  ods region;
  ods text ='Runs Data By League and Division'; ❸
  proc report data=sashelp.baseball;
    column league division nruns,(n mean min std);
    define league / group 'League';
    define division / group 'Division';
    define nruns / ''; ❹
  run;
  ods region;
  ods text='Hits Data By League and Division'; ❺
  proc report data=sashelp.baseball;
    column league division nhits,(n mean min std);
    define league / group 'League';
    define division / group 'Division';
    define nhits / ''; ❻
  run;
ods layout end;
ods powerpoint close;
```

- ❶ A null TITLE statement prevents a title from being output on the slide.
- ❷ The layout container is defined with two rows.
- ❸ An ODS TEXT= statement is used to output informative text about the table below it. It takes the place of the TITLE statement and provides more room for the table to fit on one slide.
- ❹ The label for the NRUNS variable is set to blank, which removes one of the header rows of the table.
- ❺ A ODS TEXT= statement is used to output informative text about the table below it. The text takes the place of the TITLE statement and provides more room for the table to fit on one slide.
- ❻ The label for the NHITS variable is set to blank, which removes one of the header rows of the table.

Runs Data By League and Division					
League	Division	n	mean	min	std
American	East	85	58.517647	15	28.099149
	West	90	53.188889	13	23.929623
National	East	72	49.666667	12	23.265159
	West	75	46.36	14	23.05378

Hits Data By League and Division					
League	Division	n	mean	min	std
American	East	85	113.32941	36	48.772403
	West	90	102.35556	39	41.168689
National	East	72	99.805556	31	43.805472
	West	75	96.84	32	41.427253

**Display 8. Two Tables and ODS TEXT= Placed in a Two-Row Gridded Layout**

Display 8 demonstrates that with a few small changes, all of the statistical values can be shown on one slide for both of the analysis variables. Also, though a title and a header were removed, no information was lost.

### USE MULTIPLE CONTAINERS ON ONE SLIDE

This paper has demonstrated using a layout container to move title text. It has also demonstrated putting multiple graphs and tables on one slide using either the ROWS= or COLUMNS= option in the ODS LAYOUT GRIDDED statement. This final example of ODS LAYOUT demonstrates using multiple layout containers to generate a single slide with multiple output objects.

One reason to use multiple containers for one slide is to have one output object spanning the width of the slide and placed above or below other output objects. The option ROW\_SPAN does not always work as expected in the ODS destination for PowerPoint. Instead of trying to force a spanning row, it is better to create a separate layout container for that output object.

The code in this example combines techniques already seen in previous examples. The text from PROC ODSTEXT is output into a container that is one row and one column wide. A table and a graph are output into a container that is two columns wide. The output from PROC ODSTEXT does not fill the slide, so the output from the next container automatically starts beneath it on the same slide. The size of the table and graph are adjusted to fit in the remaining space on the slide.

```

title 'Multiple Layout Containers One Slide';
ods powerpoint file='tables3.pptx' nogtitle nogfootnote;
ods layout gridded rows=1 columns=1; ❶
ods region;
proc odstext;
p 'Table Shows Total Runs and Hits for Each League';
p 'The Graph Contains One Bubble for Each Player. The Size of Each
Bubble Represents the Magnitude of the RBIs.';
run;
ods layout end;
ods layout gridded columns=2 column_widths=(47% 47%) column_gutter=1pct; ❷
ods region;
proc tabulate data=sashelp.baseball;
class league;
var nruns nhits nrbi;
tables league='', sum='Totals'*(nruns nhits)*f=comma12.;
run;


```

*code continues*

```
ods region;
proc sgplot data=sashelp.baseball;
  bubble x=nhits y=nruns size=nrbi/ group=league transparency=.3;
run;
ods layout end;
ods powerpoint close;
```

- ❶ This ODS LAYOUT GRIDDED statement creates the first layout container, which is one row wide and one column tall. The output from PROC ODSTEXT fills the container.
- ❷ This ODS LAYOUT GRIDDED statement creates a second layout container that is two columns wide. Set the column width. The COLUMN\_GUTTER option is used to decrease the space between the two output objects within this container.

Display 9 illustrates the use of multiple layout containers on a single slide.



### Display 9. Three Output Objects in Two Layout Containers

The ODSTEXT, TABULATE, and SGPLOT procedures were used to generate the output in Display 9, above. However, any three procedures could have been used in the layout. The first layout container could be filled with a table while the second layout container is filled with graphs. The important takeaway message from this example is that multiple layout containers can be placed on one slide and are sometimes necessary to get the desired output.

## THE FINAL STRETCH—USING THE LAYOUT= OPTION

As mentioned in the introduction of this paper, the ODS destination for PowerPoint has predefined layouts that are set using the LAYOUT= option in an ODS POWERPOINT statement. This paper focuses on using ODS LAYOUT within the PowerPoint destination, but it is important to be aware of the predefined layouts because they might provide what you need without having to use ODS LAYOUT. Again, both ODS LAYOUT and LAYOUT= can be used within the same presentation, but using them on the same slide is not recommended.

Three possible values for the LAYOUT= option are **TitleSlide**, **TitleAndContent**, and **TwoContent**.

The TitleSlide layout creates only title slides. Titles must be created with a P statement in PROC ODSTEXT and the number of titles is limited to two; only one footnote is allowed. This is the most straightforward way to generate a title slide and is a good first choice unless you need to change the placement of the text as was demonstrated earlier.

The TitleAndContent layout allows two titles and two footnotes. The number of output objects that fit on the slide is determined by the size of the output objects themselves. For example, two tables with three to five rows each can fit on one slide. The output objects are centered on the slide below the titles. This predefined layout stays in effect until you change it, using an ODS POWERPOINT statement with the LAYOUT= option.

The TwoContent layout places two output objects on the same slide in a two-column structure. This layout allows two titles and two footnotes. The output in a column continues until it is finished, which means it might continue across multiple slides. This predefined layout stays in effect until you change it.

The LAYOUT= option accepts a fourth value, which is `_NULL_`. This value has multiple purposes. The primary purpose is to return to the default layout state. This value should be used if you need to change from a predefined layout to the default behavior or need to create slides that use ODS LAYOUT. The second purpose of setting LAYOUT=`_NULL_` is to trigger a new slide, which it does when placed between two sections using predefined layouts. For example:

```
ods powerpoint layout=twocontent;  
proc...  
proc...  
ods powerpoint layout=_null_;  
ods powerpoint layout=titleandcontent;  
proc...
```

Predefined layouts provide an easy way to generate certain types of output structure. You should consider using them along with ODS LAYOUT when doing so would be beneficial. One advantage of predefined layouts is that they are less likely to generate warnings and suppress output from the final presentation.

## CONCLUSION

This paper demonstrated how you can use ODS LAYOUT to change the position of pieces of text, like the presentation title. ODS LAYOUT is also used to put multiple graphs on one slide or to stack tables on top of each other. Like a good pitcher and catcher, the combination of ODS LAYOUT and PowerPoint are a dynamic duo that give you the advantage that you need to win.

## REFERENCES

- Hunter, Tim. 2013. "A First Look at the ODS Destination for PowerPoint." Proceedings of the SAS Global 2013 Conference, Cary, NC: SAS Institute Inc. Available at <http://support.sas.com/resources/papers/proceedings13/041-2013.pdf>.
- Huntley, Scott. 2015. "An Insider's Guide to ODS LAYOUT Using SAS® 9.4." Proceedings of the SAS Global 2015 Conference, Cary, NC: SAS Institute Inc. Available at <http://support.sas.com/resources/papers/proceedings15/SAS1836-2015.pdf>.
- Hunter, Tim. 2016. "A Second Look at the ODS Destination for PowerPoint." Proceedings of the SAS Global 2016 Conference, Cary, NC: SAS Institute Inc.

## ACKNOWLEDGMENTS

The author is immensely grateful to Chevell Parker and Tim Hunter, both of whom contributed to this paper, and to Vivian McGee, who edited the paper.

## RECOMMENDED READING

- SAS® 9.4 Output Delivery System: User's Guide
- SAS® 9.4 Output Delivery System: Advanced Topics

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Jane Eslinger  
SAS Institute Inc.  
Cary, NC  
Email: [support@sas.com](mailto:support@sas.com)  
Web: [support.sas.com](http://support.sas.com)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.